Formal Investigation of the Extended UTxO Model

Laying the foundations for the formal verification of smart contracts

Orestis Melkonian

A thesis submitted for the Master of Science degree

Department of Information and Computing Sciences

Utrecht University

July 2019

Supervisors: Wouter Swierstra (Utrecht University)
Manuel M.T. Chakravarty (Input Output HK)

Contents

SECTION 1

Introduction

Blockchain technology has opened a whole array of interesting new applications (e.g. secure multi-party computation[?], fair protocol design fair[?], zero-knowledge proof systems[?]). Nonetheless, reasoning about the behaviour of such systems is an exceptionally hard task, mainly due to their distributed nature. Moreover, the fiscal nature of the majority of these applications requires a much higher degree of rigorousness compared to conventional IT applications, hence the need for a more formal account of their behaviour.

The advent of smart contracts (programs that run on the blockchain itself) gave rise to another source of vulnerabilities. One primary example of such a vulnerability caused by the use of smart contracts is the DAO attack[1], where a security flaw on the model of Ethereum's scripting language led to the exploitation of a venture capital fund worth 150 million dollars at the time. The solution was to create a hard fork of the Ethereum blockchain, clearly going against the decentralized spirit of cryptocurrencies. Since these (possibly Turing-complete) programs often deal with transactions of significant funds, it is of utmost importance that one can reason and ideally provide formal proofs about their behaviour in a concurrent/distributed setting.

Research Question. The aim of this thesis is to provide a mechanized formal model of an abstract distributed ledger equipped with smart contracts, in which one can begin to formally investigate the expressiveness of the extended UTxO model. Moreover, we hope to lay a foundation for a formal comparison with account-based models used in Ethereum. Put concisely, the research question posed is:

> *How much expressiveness do we gain by extending the UTxO model?*
> *Is it as expressive as the account-based model used in Ethereum?*

Overview.

- Section 2 reviews some basic definitions related to blockchain technology and introduces important literature, which will be the main subject of study throughout the development of our reasoning framework. Moreover, we give an overview of related work, putting an emphasis on existing tools based on static analysis.
- Section 3 describes the technology we will use to formally reason about the problem at hand and some key design decisions we set upfront.
- Section 4 describes the formalization of an abstract model for UTxO-based blockchain ledgers.

[1] https://en.wikipedia.org/wiki/The_DAO_(organization)

- Section 5 concerns the formalization of our second object of study, the Bitcoin Modelling Language.
- Section 7 discusses next steps for the remainder of the thesis, as well as a rough estimate on when these milestones will be completed.
- Section 8 concludes with a general overview of our contributions and reflects on the chosen methodology.

SECTION 2

Background

2.1 Distributed Ledger Technology: Blockchain

Cryptocurrencies rely on distributed ledgers, where there is no central authority managing the accounts and keeping track of the history of transactions.

One particular instance of distributed ledgers are blockchain systems, where transactions are bundled together in blocks, which are linearly connected with hashes and distributed to all peers. The blockchain system, along with a consensus protocol deciding on which competing fork of the chain is to be included, maintains an immutable distributed ledger (i.e. the history of transactions).

Validity of the transactions is tightly coupled with a consensus protocol, which makes sure peers in the network only validate well-behaved and truthful transactions and are, moreover, properly incentivized to do so.

The absence of a single central authority that has control over all assets of the participants allows for shared control of the evolution of data (in this case transactions) and generally leads to more robust and fair management of assets.

While cryptocurrencies are the major application of blockchain systems, one could easily use them for any kind of valuable asset, or even as general distributed databases.

2.2 Smart Contracts

Most blockchain systems come equipped with a scripting language, where one can write *smart contracts* that dictate how a transaction operates. A smart contract could, for instance, pose restrictions on who can redeem the output funds of a transaction.

One could view smart contracts as a replacement of legal frameworks, providing the means to conduct contractual relationships completely algorithmically.

While previous work on writing financial contracts [**?**] suggests it is fairly straightforward to write such programs embedded in a general-purpose language (in this case Haskell) and to reason about them with *equational reasoning*, it is restricted in the centralized setting and, therefore, does not suffice for our needs.

Things become much more complicated when we move to the distributed setting of a blockchain [**???**]. Hence, there is a growing need for methods and tools that will enable tractable and precise reasoning about such systems.

Numerous scripting languages have appeared recently [**?**], spanning a wide spectrum of expressiveness and complexity. While language design can impose restrictions on what a language can

express, most of these restrictions are inherited from the accounting model to which the underlying system adheres.

In the next section, we will discuss the two main forms of accounting models:

(1) **UTxO-based**: stateless models based on *unspent transaction outputs*
(2) **Account-based**: stateful models that explicitly model interaction between *user accounts*

2.3 UTxO-based: Bitcoin

The primary example of a UTxO-based blockchain is Bitcoin [**?**]. Its blockchain is a linear sequence of *blocks* of transactions, starting from the initial *genesis* block. Essentially, the blockchain acts as a public log of all transactions that have taken place, where each transaction refers to outputs of previous transactions, except for the initial *coinbase* transaction of each block. Coinbase transactions have no inputs, create new currency and reward the miner of that block with a fixed amount. Bitcoin also provides a cryptographic protocol to make sure no adversary can tamper with the transactional history, e.g. by making the creation of new blocks computationally hard and invalidating the "truthful" chain statistically impossible.

A crucial aspect of Bitcoin's design is that there are no explicit addresses included in the transactions. Rather, transaction outputs are actually program scripts, which allow someone to claim the funds by giving the proper inputs. Thus, although there are no explicit user accounts in transactions, the effective available funds of a user are all the *unspent transaction outputs* (UTxO) that he can claim (e.g. by providing a digital signature).

2.3.1 Script

In order to write such scripts in the outputs of a transaction, Bitcoin provides a low-level, Forth-like, stack-based scripting language, called Script. Script is intentionally not Turing-complete (e.g. it does not provide looping structures), in order to have more predictable behaviour. Moreover, only a very restricted set of "template" programs are considered standard, i.e. allowed to be relayed from node to node.

Script Notation. Programs in script are a linear sequence of either data values (e.g. numbers, hashes) or built-in operations (distinguished by their $\mathrm{OP\_}$ prefix).

The stack is initially considered empty and we start reading inputs from left to right. When we encounter a data item, we simply push it to the top of the stack. On encountering an operation, we pop the necessary number of arguments from the stack, apply the operation and push the result back. The evaluation function $[\![\_]\!]$ executes the given program and returns the final result at the top of the stack. For instance, adding two numbers looks like this:

$$[\![1\ 2\ \mathrm{OP\_ADD}]\!] = 3$$

P2PKH. The most frequent example of a 'standard' program in Script is the *pay-to-pubkey-hash* (P2PKH) type of scripts. Given a hash of a public key $<\mathrm{pub\#}>$, a P2PKH output carries the following script:

$$\mathrm{OP\_DUP\ OP\_HASH}\ <\mathrm{pub\#}>\ \mathrm{OP\_EQ\ OP\_CHECKSIG}$$

where OP_DUP duplicates the top element of the stack, OP_HASH replaces the top element with its hash, OP_EQ checks that the top two elements are equal, OP_CHECKSIG verifies that the top two elements are a valid pair of a digital signature of the transaction data and a public key hash.

The full script will be run when the output is claimed (i.e. used as input in a future transaction) and consists of the P2PKH script, preceded by the digital signature of the transaction by its owner and a hash of his public key. Given a digital signature <sig> and a public key hash <pub>, a transaction is valid when the execution of the script below evaluates to True.

$$\text{<sig> <pub> OP\_DUP OP\_HASH <pub\#> OP\_EQ OP\_CHECKSIG}$$

To clarify, assume a scenario where Alice want to pay Bob ₿ 10. Bob provides Alice with the cryptographic hash of his public key (<pub#>) and Alice can submit a transaction of ₿ 10 with the following output script:

$$\text{OP\_DUP OP\_HASH <pub\#> OP\_EQ OP\_CHECKSIG}$$

After that, Bob can submit another transaction that uses this output by providing the digital signature of the transaction <sig> (signed with his private key) and his public key <pub>. It is easy to see that the resulting script evaluates to True.

P2SH. A more complicated script type is *pay-to-script-hash* (P2SH), where output scripts simply authenticate against a hash of a *redeemer* script <red#>:

$$\text{OP\_HASH <red\#> OP\_EQ}$$

A redeemer script <red> resides in an input which uses the corresponding output. The following two conditions must hold for the transaction to go through:

(1) $[\![\text{<red>}]\!] = \text{True}$

(2) $[\![\text{<red> OP\_HASH <red\#> OP\_EQ}]\!] = \text{True}$

Therefore, in this case the script residing in the output is simpler, but inputs can also contain arbitrary redeemer scripts (as long as they are of a standard "template").

In this thesis, we will model scripts in a much more general, mathematical sense, so we will eschew from any further investigation of properties particular to Script.

2.3.2 The BitML Calculus

Although Bitcoin is the most widely used blockchain to date, many aspects of it are poorly documented. In general, there is a scarcity of formal models, most of which are either introductory or exploratory.

One of the most involved and mature previous work on formalizing the operation of Bitcoin is the Bitcoin Modelling Language (BitML) [**?**]. First, an idealistic *process calculus* that models Bitcoin contracts is introduced, along with a detailed small-step reduction semantics that models how contracts interact and its non-determinism accounts for the various outcomes.

The semantics consist of transitions between *configurations*, abstracting away all the cryptographic machinery and implementation details of Bitcoin. Consequently, such operational semantics allow one to reason about the concurrent behaviour of the contracts in a *symbolic* setting.

The authors then provide a compiler from BitML contracts to 'standard' Bitcoin transactions, proven correct via a correspondence between the symbolic model and the computational model operating on the Bitcoin blockchain. We will return for a more formal treatment of BitML in Section ??.

2.3.3 *Extended UTxO*

In this work, we will consider the version of the UTxO model used by IOHK's Cardano blockchain[2]. In contrast to Bitcoin's *proof-of-work* consensus protocol [?], Cardano's *Ouroboros* protocol [?] is *proof-of-stake*. This, however, does not concern our study of the abstract accounting model, thus we refrain from formally modelling and comparing different consensus techniques.

The actual extension we care about is the inclusion of *data scripts* in transaction outputs, which essentially provide the validation script in the corresponding input with additional information of an arbitrary type.

This extension of the UTxO model has already been implemented[3], but only informally documented[4]. The reason to extend the UTxO model with data scripts is to bring more expressive power to UTxO-based blockchains, hopefully bringing it on par with Ethereum's account-based scripting model (see Section 2.4).

However, there is no formal argument to support this claim, and it is the goal of this thesis to provide the first formal investigation of the expressiveness introduced by this extension.

2.4 Account-based: Ethereum

On the other side of the spectrum, lies the second biggest cryptocurrency today, Ethereum [?]. In contrast to UTxO-based systems, Ethereum has a built-in notion of user addresses and operates on a stateful accounting model. It goes even further to distinguish *human accounts* (controlled by a public-private key pair) from *contract accounts* (controlled by some EVM code).

This added expressiveness is also reflected in the quasi-Turing-complete low-level stack-based bytecode language in which contract code is written, namely the *Ethereum Virtual Machine* (EVM). EVM is mostly designed as a target, to which other high-level user-friendly languages will compile to.

Solidity. The most widely adopted language that targets the EVM is *Solidity*, whose high-level object-oriented design makes writing common contract use-cases (e.g. crowdfunding campaigns, auctions) rather straightforward.

One of Solidity's most distinguishing features is the concept of a contract's *gas*; a limit to the amount of computational steps a contract can perform. At the time of the creation of a transaction, its owner specifies a certain amount of gas the contract can consume and pays a transaction fee proportional to it. In case of complete depletion (i.e. all gas has been consumed before the contract finishes its execution), all global state changes are reverted as if the contract had never been

[2] www.cardano.org

[3] https://github.com/input-output-hk/plutus/tree/master/wallet-api/src/Ledger

[4] https://github.com/input-output-hk/plutus/blob/master/docs/extended-utxo/README.md

run. This is a necessary ingredient for smart contract languages that provide arbitrary looping behaviour, since non-termination of the validation phase is certainly undesirable.

If time permits, we will initially provide a formal justification of Solidity and proceed to formally compare the extended UTxO model against it. Since Solidity is a fully-fledged programming language with lots of features (e.g. static typing, inheritance, libraries, user-defined types), it makes sense to restrict our formal study to a compact subset of Solidity that is easy to reason about. This is the approach also taken in Featherweight Java [**?**]; a subset of Java that omits complex features such as reflection, in favour of easier behavioural reasoning and a more formal investigation of its semantics. In the same vein, we will try to introduce a lightweight version of Solidity, which we will refer to as *Featherweight Solidity*.

SECTION 3

Methodology

3.1 Scope

At this point, we have to stress the fact that we are not aiming for a formalization of a fully-fledged blockchain system with all its bells and whistles, but rather focus on the underlying accounting model. Therefore, we will omit details concerning cryptographic operations and aspects of the actual implementation of such a system. Instead, we will work on an abstract layer that postulates the well-behavedness of these subcomponents, which will hopefully lend itself to more tractable reasoning and give us a clear overview of the essence of the problem.

Restricting the scope of our attempt is also motivated from the fact that individual components such as cryptographic protocols are orthogonal to the functionality we study here. This lack of tight cohesion between the components of the system allows one to safely factor each one out and formalize it independently.

It is important to note that this is not always the case for every domain. A prominent example of this are operating systems, which consist of intricately linked subcomponents (e.g. drivers, memory modules), thus making it impossible to trivially divide the overall proof into small independent ones. In order to overcome this complexity burden, one has to invent novel ways of modular proof mechanization, as exemplified by *CertiKOS* [**?**], a formally verified concurrent OS.

3.2 Proof Mechanization

Fortunately, the sub-components of the system we are examining are not no interdependent, thus lending themselves to separate treatment. Nonetheless, the complexity of the sub-system we care about is still high and requires rigorous investigation. Therefore, we choose to conduct our formal study in a mechanized manner, i.e. using a proof assistant along the way and formalizing all results in Type Theory. Proof mechanization will allow us to discover edge cases and increase the confidence of the model under investigation.

3.3 Agda

As our proof development vehicle, we choose Agda [**?**], a dependently-typed total functional language similar to Haskell [**?**].

Agda embraces the *Curry-Howard correspondence*, which states that types are isomorphic to statements in (intuitionistic) logic and their programs correspond to the proofs of these statements [**?**]. Through its unicode-based *mixfix* notational system, one can easily translate a mathematical theorem into a valid Agda type. Moreover, programs and proofs share the same structure, e.g. induction in the proof manifests itself as recursion in the program.

While Agda is not ideal for large software development, its flexible notation and elegant design is suitable for rapid prototyping of new ideas and exploratory purposes. We do not expect to hit such problems, since we will stay on a fairly abstract level which postulates cryptographic operations and other implementation details.

Limitation. The main limitation of Agda lies in its lack of a proper proof automation system. While there has been work on providing Agda with such capabilities [**?**], it requires moving to a meta-programming mindset which would be an additional programming hindrance.

A reasonable alternative would be to use Coq [**?**], which provides a pragmatic scripting language for programming *tactics*, i.e. programs that work on proof contexts and can generate new subgoals. This approach to proof mechanization has, however, been criticized for widening the gap between informal proofs and programs written in a proof assistant. This clearly goes against the aforementioned principle of *proofs-as-programs*.

3.4 The IOHK approach

At this point, we would like to mention the specific approach taken by IOHK[5]. In contrast to numerous other companies currently creating cryptocurrencies, its main focus is on provably correct protocols with a strong focus on peer-reviewing and robust implementations, rather than fast delivery of results. This is evidenced by the choice of programming languages (Agda/Coq/Haskell/Scala) – all functional programming languages with rich type systems – and the use of *property-based testing* [**?**] for the production code.

IOHK's distinct feature is that it advocates a more rigorous development pipeline; ideas are initially worked on paper by pure academics, which create fertile ground for starting formal verification in Agda/Coq for more confident results, which result in a prototype/reference implementation in Haskell, which informs the production code-base (also written in Haskell) on the properties that should be tested.

Since this thesis is done in close collaboration with IOHK, it is situated on the second step of aforementioned pipeline; while there has been work on writing papers about the extended UTxO model along with the actual implementation in Haskell, there is still no complete and mechanized account of its properties.

3.5 Functional Programming Principles

One last important manifestation of the functional programming principles behind IOHK is the choice of a UTxO-based cryptocurrency itself.

On the one hand, one can view a UTxO ledger as a dataflow diagram, whose nodes are the submitted transactions and edges represent links between transaction inputs and outputs. On the

[5] https://iohk.io/

other hand, account-based ledgers rely on a global state and transaction have a much more complicated specification.

The key point here is that UTxO-based transaction are just pure mathematical functions, which are much more straightforward to model and reason about. Coming back to the principles of functional programming, one could contrast this with the difference between functional and imperative programs. One can use *equational reasoning* for functional programs, due to their *referential transparency*, while this is not possible for imperative programs that contain side-effectful commands. Therefore, we hope that these principles will be reflected in the proof process itself; one would reason about purely functional UTxO-based ledgers in a compositional manner.

This section gives an overview of the progress made so far in the on-going Agda formalization of the two main subjects of study, namely the Extended UTxO model and the BitML calculus. For the sake of brevity, we refrain from showing the full Agda code along with the complete proofs, but rather provide the most important datatypes and formalized results and explain crucial design choices we made along the way. Furthermore, we will omit notational burden imposed by technicalities particular to Agda, such as *universe polymorphism* and *proof irrelevance*.

SECTION 4

Formal Model I: Extended UTxO

We now set out to model the accounting model of a UTxO-based ledger. We will provide a inherently-typed model of transactions and ledgers; this gives rise to a notion of *weakening* of available addresses, which we formalize. Moreover, we showcase the reasoning abilities of our model by giving an example of a correct-by-construction ledger. All code is publicly available on Github[6].

We start with the basic types, keeping them abstract since we do not care about details arising from the encoding in an actual implementation:

```
postulate
  Address : Set
  Value : Set
  ₿ : ℕ → Value
```

We assume there are types representing addresses and bitcoin values, but also require the ability to construct a value out of a natural number. In the examples that follow, we assume the simplest representation, where both types are the natural numbers.

There is also the notion of the *state* of a ledger, which will be provided to transaction scripts and allow them to have stateful behaviour for more complicated schemes (e.g. imposing time constraints).

```
record State : Set where
  field height : ℕ
        ⋮
```

The state components have not been finalized yet, but can easily be extended later when we actually investigate examples with expressive scripts that make use of state information, such as the current length of the ledger (*height*).

As mentioned previously, we will not dive into the verification of the cryptological components of the model, hence we postulate an *irreversible* hashing function which, given any value of any type, gives back an address (i.e. a natural number) and is furthermore injective (i.e. it is highly unlikely for two different values to have the same hash).

[6] https://github.com/omelkonian/formal-utxo

```
postulate
  _♯ : ∀ {A : Set} → A → Address
  ♯-injective : ∀ {x y : A} → x ♯ ≡ y ♯ → x ≡ y
```

4.1 Transactions

In order to model transactions that are part of a distributed ledger, we need to first define transaction *inputs* and *outputs*.

```
record TxOutputRef : Set where
  constructor _@_
  field id    : Address
        index : ℕ

record TxInput {R D : Set} : Set where
  field outputRef : TxOutputRef
        redeemer  : State → R
        validator : State → Value → R → D → Bool
```

Output references consist of the address that a transaction hashes to, as well as the index in this transaction's list of outputs. *Transaction inputs* refer to some previous output in the ledger, but also contain two types of scripts. The *redeemer* provides evidence of authorization to spend the output. The *validator* then checks whether this is so, having access to the current state of the ledger, the bitcoin output and data provided by the redeemer and the *data script* (residing in outputs). It is also noteworthy that we immediately model scripts by their *denotational semantics*, omitting unnecessary details relating to concrete syntax, lexing and parsing.

Transaction outputs send a bitcoin amount to a particular address, which either corresponds to a public key hash of a blockchain participant (P2PKH) or a hash of a next transaction's script (P2SH). Here, we opt to embrace the *inherently-typed* philosophy of Agda and model available addresses as module parameters. That is, we package the following definitions in a module with such a parameter, as shown below:

```
module UTxO (addresses : List Address) where

record TxOutput {D : Set} : Set where
  field value      : Value
        address    : Index addresses
        dataScript : State → D

record Tx : Set where
  field inputs  : Set⟨ TxInput ⟩
        outputs : List TxOutput
        forge   : Value
        fee     : Value
```

```
Ledger : Set
Ledger = List Tx
```

Transaction outputs consist of a bitcoin amount and the address (out of the available ones) this amount is sent to, as well as the data script, which provides extra information to the aforementioned validator and allows for more expressive schemes. Investigating exactly the extent of this expressiveness is one of the main goals of this thesis.

For a transaction to be submitted, one has to check that each input can actually spend the output it refers to. At this point of interaction, one must combine all scripts, as shown below:

```
runValidation : (i : TxInput) → (o : TxOutput) → D i ≡ D o → State → Bool
runValidation i o refl st = validator i st (value o) (redeemer i st) (dataScript o st)
```

Note that the intermediate types carried by the respective input and output must align, evidenced by the equality proof that is required as an argument.

4.2 Unspent Transaction Outputs

With the basic modelling of a ledger and its transaction in place, it is fairly straightforward to inductively define the calculation of a ledger's unspent transaction outputs:

```
unspentOutputs : Ledger → Set⟨ TxOutputRef ⟩
unspentOutputs [] = ∅
unspentOutputs (tx :: txs) = (unspentOutputs txs \ spentOutputsTx tx) ∪ unspentOutputsTx tx
  where
    spentOutputsTx , unspentOutputsTx : Tx → Set⟨ TxOutputRef ⟩
    spentOutputsTx      = (outputRef ⟨$⟩_) ∘ inputs
    unspentOutputsTx tx = ((tx ♯) @ _)⟨$⟩(indices (outputs tx))
```

4.3 Validity of Transactions

In order to submit a transaction, one has to make sure it is valid with respect to the current ledger. We model validity as a record indexed by the transaction to be submitted and the current ledger:

```
record IsValidTx (tx : Tx) (l : Ledger) : Set where
  field
    validTxRefs :
      ∀ i → i ∈ inputs tx →
        Any (λ t → t ♯ ≡ id (outputRef i)) l

    validOutputIndices :
      ∀ i → (i∈ : i ∈ inputs tx) →
        index (outputRef i) <
```

```
        length (outputs (lookupTx l (outputRef i) (validTxRefs i i∈)))

validOutputRefs :
  ∀ i → i ∈ inputs tx →
    outputRef i ∈ unspentOutputs l

validDataScriptTypes :
  ∀ i → (i∈ : i ∈ inputs tx) →
    D i ≡ D (lookupOutput l (outputRef i) (validTxRefs i i∈) (validOutputIndices i i∈))
```

```
preservesValues :
  forge tx + sum (mapWith ∈ (inputs tx) λ {i} i∈ →
    lookupValue l i (validTxRefs i i∈) (validOutputIndices i i∈))
    ≡
  fee tx + sum (value⟨$⟩outputs tx)

noDoubleSpending :
  noDuplicates (outputRef⟨$⟩inputs tx)

allInputsValidate :
  ∀ i → (i∈ : i ∈ inputs tx) →
    let out : TxOutput
        out = lookupOutput l (outputRef i) (validTxRefs i i∈) (validOutputIndices i i∈)
    in ∀ (st : State) →
         T (runValidation i out (validDataScriptTypes i i∈) st)

validateValidHashes :
  ∀ i → (i∈ : i ∈ inputs tx) →
    let out : TxOutput
        out = lookupOutput l (outputRef i) (validTxRefs i i∈) (validOutputIndices i i∈)
    in toℕ (address out) ≡ (validator i) ♯
```

The first four conditions make sure the transaction references and types are well-formed, namely that inputs refer to actual transactions (*validTxRefs*, *validOutputIndices*) which are unspent so far (*validOutputRefs*), but also that intermediate types used in interacting inputs and outputs align (*validDataScriptTypes*).

The last four validation conditions are more interesting, as they ascertain the validity of the submitted transaction, namely that the bitcoin values sum up properly (*preservesValues*), no output is

spent twice (*noDoubleSpending*), validation succeeds for each input-output pair (*allInputsValidate*) and outputs hash to the hash of their corresponding validator script (*validateValidHashes*).

The definitions of lookup functions are omitted, as they are uninteresting. The only important design choice is that, instead of modelling lookups as partial functions (i.e. returning *Maybe*), they require a membership proof as an argument moving the responsibility to the caller (as evidenced by their usage in the validity conditions).

Type-safe interface. Users should only have a type-safe interface to construct ledgers, where each time a transaction is submitted along with the proof that it is valid with respect to the ledger constructed thus far. We provide such an interface with a proof-carrying variant of the standard list construction:

```
data ValidLedger : Ledger → Set where
  • : ValidLedger []
  _⊕_⊣_ : ValidLedger l
        → (tx : Tx)
        → IsValidTx tx l
        → ValidLedger (tx :: l)
```

4.4 Decision Procedure

Intrinsically-typed ledgers are correct-by-construction, but this does not come for free; we now need to provide substantial proofs alongside each time we submit a new transaction.

To make the proof process more ergonomic for the user of the framework, we prove that all involved propositions appearing in the *IsValidTx* record are *decidable*, thus defining a decision procedure for closed formulas that do not contain any free variable. This process is commonly referred to as *proof-by-reflection* [Van Der Walt and Swierstra 2012].

Most operations already come with a decidable counterpart, e.g. $\_ < \_$ can be decided by $\_ <? \_$ that exists in Agda's standard library. Therefore, what we are essentially doing is copy the initial propositions and replace such operators with their decision procedures. Decidability is captured by the *Dec* datatype, ensuring that we can answer a yes/no question over the enclosed proposition:

```
data Dec (P : Set) : Set where
  yes : (p : P)  → Dec P
  no  : (¬p : ¬P) → Dec P
```

Having a proof of decidability means we can replace a proof of proposition P with a simple call to $toWitness\ \{Q = P?\}\ tt$, where $P?$ is the decidable counterpart of P.

```
True : Dec P → Set
True (yes _) = ⊤
True (no _)  = ⊥
```

```
toWitness : {Q : Dec P} → True Q → P
toWitness {Q = yes p} _ = p
toWitness {Q = no _}  ()
```

For this to compute though, the decided formula needs to be *closed*, meaning it does not contain any variables. One could even go beyond closed formulas by utilizing Agda's recent *metaprogramming* facilities (macros), but this is outside of the scope of this thesis.

But what about universal quantification? We certainly know that it is not possible to decide on an arbitrary quantified proposition. Hopefully, all our uses of the ∀ operator later constrain the quantified argument to be an element of a list. Therefore, we can define a specific decidable variant of this format:

```
∀? : (xs : List A)
   → {P : (x : A) (x∈ : x ∈ xs) → Set}
   → (∀ x → (x∈ : x ∈ xs) → Dec (P x x∈))
   → Dec (∀ x x∈ → P x x∈)
∀? []       P? = yes λ _ ()
∀? (x :: xs) P? with ∀? xs (λ x′ x∈ → P? x′ (there x∈))
... | no ¬p     = no λ p → ¬p (λ x′ x∈ → p x′ (there x∈))
... | yes p′    with P? x (here refl)
... | no ¬p     = no λ p → ¬p (p x (here refl))
... | yes p     = yes λ {x′ (here refl) → p
                        ; x′ (there x∈) → p′ x′ x∈}
```

Finally, we are ready to provide a decision procedure for each validity condition using the aforementioned operators for quantification and the decidable counterparts for the standard operators we use. Below we give an example for the *validOutputRefs* condition:

```
validOutputRefs? : ∀ (tx : Tx) (l : Ledger)
  → Dec (∀ i → i ∈ inputs tx → outputRef i ∈ unspentOutputs l)
validOutputRefs? tx l =
  ∀? (inputs tx) λ i _ →
    outputRef i∈? unspentOutputs l
```

In Section 4.9 we give an example construction of a valid ledger and demonstrate that our decision procedure discharges all proof obligations with calls to *toWitness*.

4.5 Weakening Lemma

We have defined everything with respect to a fixed set of available addresses, but it would make sense to be able to include additional addresses without losing the validity of the ledger constructed thus far.

In order to do, we need to first expose the basic datatypes from inside the module, introducing their *primed* version which takes the corresponding module parameter as an index:

```
Ledger′ : List Address → Set
Ledger′ as = Ledger
  where open import UTxO as
⋮
```

We can now precisely define what it means to weaken an address space; the only necessary ingredient is a *hash-preserving injection* from a smaller address space $\mathbb{A}$ to a larger address space $\mathbb{B}$:

```
module Weakening
  (𝔸 : Set) (_ ♯ᵃ : Hash 𝔸) (_ ≟ᵃ _ : Decidable {A = 𝔸} _≡_)
  (𝔹 : Set) (_ ♯ᵇ : Hash 𝔹) (_ ≟ᵇ _ : Decidable {A = 𝔹} _≡_)
  (A ↪ B : 𝔸 , _ ♯ᵃ ↪ 𝔹 , _ ♯ᵇ)
  where

  import UTxO.Validity 𝔸 _ ♯ᵃ _ ≟ᵃ _ as A
  open import UTxO.Validity 𝔹 _ ♯ᵇ _ ≟ᵇ _ as B

  weakenTxOutput : A.TxOutput → B.TxOutput
  weakenTxOutput out = out { address = A ↪ B⟨$⟩(address out) }

  weakenTx : A.Tx → B.Tx
  weakenTx tx = tx { outputs = map weakenTxOutput (outputs tx) }

  weakenLedger : A.Ledger → B.Ledger
  weakenLedger = map weakenTx
```

Notice also that the only place where weakening takes place are transaction outputs, since all other components do not depend on the available address space.

With the weakening properly defined, we can finally prove the *weakening lemma* for the available address space:

```
weakening : ∀ {tx : A.Tx} {l : A.Ledger}
  → A.IsValidTx tx l
  ──────────────────────────────────
  → B.IsValidTx (weakenTx tx) (weakenLedger l)
weakening = ...
```

The weakening lemma states that the validity of a transaction with respect to a ledger is preserved if we choose to weaken the available address space, which we estimate to be useful when we later prove more intricate properties of the extended UTxO model.

One practical use-case for weakening is moving from a bit representation of addresses to one with more available bits (e.g. 32-bit to 64-bit conversion). This, of course, preserves hashes since the numeric equivalent of the converted addresses will be the same. For instance, as we come closer to the quantum computing age, addresses will have to transition to other encryption schemes involving many more bits[7]. Since we allow the flexibility for arbitrary injective functions, our weakening result will hopefully prove resilient to such scenarios.

4.6 Combining

Ideally, one would wish for a modular reasoning process, where it is possible to examine subsets of unrelated transactions in a compositional manner. This has to be done in a constrained manner, since we need to preserve the proof of validity when combining two ledgers l and l'.

First of all, the ledgers should not share any transactions with each other: *Disjoint* $l\ l'$. Secondly, the resulting ledger l'' will be some interleaving of these two: *Interleaving* $l\ l'\ l''$. These conditions are actually sufficient to preserve all validity conditions, except *allInputsValidate*. The issue arises from the dependence of validation results on the current state of the ledger, which is given as argument to each validation script. To remedy this, we further require that the new state, corresponding to a particular interleaving, does not break previous validation results:

```
PreserveValidations : (l : Ledger) (l″ : Ledger) → Interleaving l _ l″ → Set
PreserveValidations l₀ _ inter =
  ∀ tx → (p : tx ∈ l₀) →
    let l  = ∈-tail p
        l″ = ∈-tail (interleave ⊆ inter p)
    in ∀ {ptx i out vds} → runValidation ptx i out vds (getState l″)
                         ≡ runValidation ptx i out vds (getState l)
```

Putting all conditions together, we are now ready to formulate a *combining* operation for valid ledgers:

```
_ ↔ _ ⊣ _ : ∀ {l l′ l″ : Ledger}
  → ValidLedger l
  → ValidLedger l′
  → Σ[ i ∈ Interleaving l l′ l″ ]
  × Disjoint l l′
  × PreserveValidations l l″ i
  × PreserveValidations l′ l″ (swap i)
```

[7] It is believed that even 2048-bit keys will become vulnerable to rapid decryption from quantum computers.

$$\rightarrow ValidLedger\ l''$$

The proof inductively proves validity of each transaction in the interleaved ledger, essentially reusing the validity proofs of the ledger constituents.

It is important to notice a useful interplay between weakening and combining: if we wish to combine ledgers that use different addresses, we can now just apply weakening first and then combine in a type-safe manner.

4.7 Extension I: Data Scripts

The *dataScript* field in transaction outputs does not appear in the original abstract UTxO model [Zahnentferner 2018], but is available in the extended version of the UTxO model used in the Cardano blockchain [eut 2019]. This addition raises the expressive level of UTxO-based transaction, since it is now possible to simulate stateful behaviour, passing around state in the data scripts (i.e. $D = State$).

This technique is successfully employed in *Marlowe*, a DSL for financial contracts that compiles down to eUTxO transactions [Seijas and Thompson 2018]. Marlowe is accompanied by a simple small-step semantics, i.e. a state transition system. Using data scripts, compilation is rather straightforward since we can pass around the state of the semantics in the data scripts.

4.8 Extension II: Multi-currency

Many major blockchain systems today support the creation of secondary cryptocurrencies, which are independent of the main currency. In Bitcoin, for instance, *colored coins* allow transactions to assign additional meaning to their outputs (e.g. each coin could correspond to a real-world asset, such as company shares) [Rosenfeld 2012].

This approach, however, has the disadvantage of larger transactions and less efficient processing. One could instead bake the multi-currency feature into the base system, mitigating the need for larger transactions and slow processing. Building on the abstract UTxO model, there are current research efforts on a general framework that provides mechanisms to establish and enforce monetary policies for multiple currencies [Zahnentferner 2019].

Fortunately, the extensions proposed by the multi-currency are orthogonal to the formalization so far. In order to accommodate built-in support for user-defined currencies, we need to generalize the type of $Value$ from quantities ($\mathbb{N}$) to maps from *currency identifiers* to quantities.

Thankfully, the value operations used in our validity conditions could be lifted to any *commutative group*[8]. Hence, refactoring the validity conditions consists of merely replacing numeric addition with a point-wise addition on maps $\_ +^{c} \_$.

At the user-level, we define these value maps as a simple list of key-value pairs:

$$Value = List\ (Hash \times \mathbb{N})$$

[8] Actually, we only ever *add* values, but inverses could be used to *reduce* a currency supply.

Note that currency identifiers are not strings, but script hashes. We will justify this decision when we talk about the way *monetary policies* are enforced; each currency comes with a certain scheme of allowing or refusing forging of new funds.

We also provide the adding operation, internally using proper maps implemented on AVL trees [9]:

```
open import Data.AVL ℕ-strictTotalOrder

CurrencyMap = Tree (MkValue (λ _ → ℕ) (subst (λ _ → ℕ)))

_+ᶜ_ : Value → Value → Value
c +ᶜ c′ = toList (foldl go (fromList c) c′)
  where
    go : CurrencyMap → (ℕ × ℕ) → CurrencyMap
    go cur (currency , value) = insertWith currency ((_+ value) ∘ fromMaybe 0) cur

sumᶜ : List Value → Value
sumᶜ = foldl _+ᶜ_ []
```

While the multi-currency paper defines a new type of transaction *CurrencyTx* for creating funds, we follow a more lightweight approach, currently employed in the Cardano blockchain [mul 2019]. This proposal mitigates the need for a new type of transaction and a global registry via a clever use of validator scripts: monetary policies reside in the validator script of the transactional inputs and currency identifiers are just the hashes of those scripts. When one needs to forge a particular currency, two transactions must be submitted: the first only carrying the monetary policy in its output and the second consuming it and forging the desired quantity.

In order to ascertain that forging transactions always follow this scheme, we need to extend our validity record with yet another condition:

```
record IsValidTx (tx : Tx) (l : Ledger) : Set where
  ...
  forging :
    ∀ c → c ∈ keys (forge tx) →
      ∃[ i ] ∃λ (i∈ : i ∈ inputs tx) →
        let out = lookupOutput l (outputRef i) (validTxRefs i i∈) (validOutputIndices i i∈)
        in (address out) ♯ ≡ c
```

The rest of the conditions are the same, modulo the replacement of $\_ + \_$ with $\_ +^{c} \_$ and *sum* with sum^{c}.

[9] https://github.com/agda/agda-stdlib/blob/master/src/Data/AVL.agda

This is actually the first and only validation condition to contain an existential quantification, which poses some issues with our decision procedure for validity. To tackle this, we follow a similar approach to the treatment of universal quantification in Section 4.4:

```
∃?  :  (xs : List A)
    → {P : (x : A) (x∈ : x ∈ xs) → Set ℓ′ }
    → (∀ x → (x∈ : x ∈ xs) → Dec (P x x∈))
    → Dec (∃[ x ]∃λ (x∈ : x ∈ xs) → P x x∈)
∃? [] P?              = no λ {(x , () , p)}
∃? (x :: xs) P?       with P? x (here refl)
... | yes kp          = yes (x , here refl , p)
... | no¬p            with ∃? xs (λ x′ x∈ → P? x′ (there x∈))
... | yes (x′ , x∈ , p) = yes (x′ , there x∈ , p)
... | no¬pp           = no λ {(x′ , here refl , p) → ¬p p
                        ; (x′ , there x∈ , p) → ¬pp (x′ , x∈ , p)}
```

Now it is straightforward to give a proof of decidability for *forging*:

```
forging? : ∀ (tx : Tx) (l : Ledger)
  → (v₁ : ∀ i → i ∈ inputs tx → Any (λ t → t ♯ ≡ id (outputRef i)) l)
  → (v₂ : ∀ i → (i∈ : i ∈ inputs tx) →
        index (outputRef i) < length (outputs (lookupTx l (outputRef i) (v₁ i i∈))))
  → Dec (∀ c → c ∈ keys (forge tx) →
          ∃[ i ] ∃λ (i∈ : i ∈ inputs tx) →
            let out = lookupOutput l (outputRef i) (v₁ i i∈) (v₂ i i∈)
            in (address out) ♯ ≡ c)
forging? tx l v₁ v₂ =
  ∀? (keys (forge tx)) λ c _ →
    ∃? (inputs tx) λ i i∈ →
      let out = lookupOutput l (outputRef i) (v₁ i i∈) (v₂ i i∈)
      in (address out) ♯ ≟ c
```

4.9 Example

To showcase how we can use our model to construct *correct-by-construction* ledgers, let us revisit the example ledger presented in the Chimeric Ledgers paper [Zahnentferner and HK 2018].

Any blockchain can be visually represented as a *directed acyclic graph* (DAG), with transactions as nodes and input-output pairs as edges, as shown in Figure 1. The six transactions $t_1 \ \ldots \ t_6$ are self-explanatory, each containing a forge and fee value. Notice the special transaction c_0, which



Fig. 1. Example ledger with six transactions (unspent outputs are coloured in red)

enforces the monetary policy of currency ₿ in its outputs (colored in green); the two forging transactions t_1 and t_4 consume these outputs as requested by the validity condition for *forging*. Lastly, there is a single unspent output (coloured in red), namely the single output of t_6.

First, we need to set things up by declaring the list of available addresses and opening our module with this parameter. For brevity, we view addresses immediately as hashes:

```
Address : Set
Address = ℕ

1ᵃ , 2ᵃ , 3ᵃ , ₿ᵃ : Address
1ᵃ = 111  -- first address
2ᵃ = 222  -- second address
3ᵃ = 333  -- third address
₿ᵃ = 1234 -- BIT identifier

open import UTxO Address (λ x → x) _≟_
```

It is also convenient to define some smart constructors up-front:

```
₿-validator : State → ... → Bool
₿-validator (record {height = h}) _ _ _ _ = h ≡ᵇ 1 ∨ h ≡ᵇ 4

mkValidator : TxOutputRef → (State → Value → PendingTx → (ℕ × ℕ) → ℕ → Bool)
mkValidator tin _ _ _ tin' _ = (id tin ≡ᵇ proj₁ tin') ∧ (index tin ≡ᵇ proj₂ tin')

₿_ : ℕ → Value
₿ v = [(₿ᵃ , v)]
```

```
withScripts : TxOutputRef → TxInput
withScripts tin = record { outputRef = tin
                         ; redeemer  = λ _ → id tin , index tin
                         ; validator = mkValidator tin
                         }

withPolicy : TxOutputRef → TxInput
withPolicy tin = record { outputRef = tin
                        ; redeemer = λ _ → tt
                        ; validator = ₿-validator
                        }

_@_ : Value → Index addresses → TxOutput
v @ addr = record { value = v; address = addr; dataScript = λ _ → tt }
```

₿-*validator* models a monetary policy that allows forging only at ledger height 1 and 4; *mkValidator* is a script that only validates against the given output reference; ₿_ creates singleton currency maps for our currency BIT; *withScripts* and *withPolicy* wrap an output reference with the appropriate scripts; _@_ creates outputs that do not utilize the data script.

We can then proceed to define the individual transactions defined in Figure 1; the first sub-index of each variable refers to the order the transaction are submitted, while the second sub-index refers to which output of the given transaction we select:

```
c₀ , t₁ , t₂ , t₃ , t₄ , t₅ , t₆ : Tx
c₀ = record { inputs = [ ]
            ; outputs = ₿ 0 @ (₿-validator ♯) :: ₿ 0 @ (₿-validator ♯) :: [ ]
            ; forge  = ₿ 0
            ; fee    = ₿ 0
            }
t₁ = record { inputs  = [ withPolicy c₀₀ ]
            ; outputs = [ ₿ 1000 @ 0 ]
            ; forge   = ₿ 1000
            ; fee     = ₿ 0
            }
t₂ = record { inputs  = [ withScripts t₁₀ ]
            ; outputs = ₿ 800 @ 1 :: ₿ 200 @ 0 :: [ ]
            ; forge   = ₿ 0
            ; fee     = ₿ 0
            }
```

```
t₃ = record { inputs  = [ withScripts t₂₁ ]
            ; outputs = [ ₿ 199 @ 2 ]
            ; forge   = ₿ 0
            ; fee     = ₿ 1
            }
t₄ = record { inputs  = withScripts t₃₀ :: withPolicy c₀₁ :: [ ]
            ; outputs = [ ₿ 207 @ 1 ]
            ; forge   = ₿ 10
            ; fee     = ₿ 2
            }
t₅ = record { inputs  = withScripts t₂₀ :: withScripts t₄₀ :: [ ]
            ; outputs = ₿ 500 @ 1 :: ₿ 500 @ 2 :: [ ]
            ; forge   = ₿ 0
            ; fee     = ₿ 7
            }
t₆ = record { inputs  = withScripts t₅₀ :: withScripts t₅₁ :: [ ]
            ; outputs = [ ₿ 999 @ 2 ]
            ; forge   = ₿ 0
            ; fee     = ₿ 1
            }
```

In order for terms involving the *postulated* hash function _♯ to compute, we use Agda's experimental feature for user-supplied *rewrite rules*:

```
{-# OPTIONS --rewriting #-}
postulate
  eq₀  : ₿-validator        ♯ ≡ ₿ ᵃ
  eq₁₀ : (mkValidator t₁₀) ♯ ≡ 1 ᵃ
  ⋮
  eq₆₀ : (mkValidator t₆₀) ♯ ≡ 3 ᵃ

{-# BUILTIN REWRITE _≡_ #-}
{-# REWRITE eq₀ , eq₁₀ , ... , eq₆₀ #-}
```

Below we give a correct-by-construction ledger containing all transactions:

```
ex-ledger : ValidLedger (t₆ :: t₅ :: t₄ :: t₃ :: t₂ :: t₁ :: c₀ :: [ ])
ex-ledger = ∙ c₀ ⊣ record { ... }
            ⊕ t₁ ⊣ record { validTxRefs = toWitness {Q = validTxRefs? t₁ l₀} tt
```

```
                    ; validOutputIndices    = toWitness {Q = validOutputIndices? ...} tt
                    ; validOutputRefs       = toWitness {Q = validOutputRef? ...} tt
                    ; validDataScriptTypes  = toWitness {Q = validDataScriptTypes? ...} tt
                    ; preservesValues       = toWitness {Q = preservesValues? ...} tt
                    ; noDoubleSpending      = toWitness {Q = noDoubleSpending? ...} tt
                    ; allInputsValidate     = toWitness {Q = allInputsValidate? ...} tt
                    ; validateValidHashes   = toWitness {Q = validateValidHashes? ...} tt
                    ; forging               = toWitness {Q = forging? ...} tt
                    }
  ⊕ t₂ ⊣ record {...}
  ⋮
  ⊕ t₆ ⊣ record {...}
```

First, it is trivial to verify that the only unspent transaction output of our ledger is the output of the last transaction t_6, as demonstrated below:

```
utxo : list (unspentOutputs ex-ledger) ≡ [ t₆₀ ]
utxo = refl
```

Most importantly, notice that no manual proving is necessary, since our decision procedure discharges all validity proofs. In the next release of Agda, it will be possible to even omit the manual calls to the decision procedure (via *toWitness*), by declaring the proof of validity as an implicit *tactic argument*[10].

This machinery allows us to define a compile-time macro for each validity condition that works on the corresponding goal type, and *statically* calls the decision procedure of this condition to extract a proof and fill the required implicit argument. As an example, we give a sketch of the macro for the *validTxRefs* condition below:

```
pattern vtx i i∈ tx t l =
  `λ i : `TxInput ⇒
    `λ i∈ : #0 ` ∈ (`inputs t) ⇒
      `Any (`λ tx ⇒ #0 ` ♯ ` ≡ `id `outputRef #2) l

macro
  validTxRefsM : Term → TC ⊤
  validTxRefsM hole = do
    goal ← inferType hole
    case goal of λ
      {(vtx _ _ _ t l) →
        t′ ← unquoteTC t
```

[10] https://agda.readthedocs.io/en/latest/language/implicit-arguments.html#tactic-arguments

```
    l′ ←unquoteTC l
    case validTxRefs?t′ l′ of λ
      { (yes p) → quoteTC p >>= unify hole
      ; (no _)  → typeError [ strErr "validity condition does not hold" ]
      }
  ; t → typeError [ strErr "wrong type of goal" ]
  }
```

We first define a pattern to capture the validity condition in AST form; Agda provides a *reflection* mechanism[11], that defines Agda's language constructs as regular Agda datatypes. Note the use of *quoted* expressions in the definition of the *vtx* pattern, which also uses De Bruijn indices for variables bound in λ-abstractions.

Then, we define the macro as a *metaprogram* running in the type-checking monad *TC*. After pattern matching on the goal type and making sure it has the expected form, we run the decision procedure, in this case *validTxRefs?*. If the computation reached a positive answer, we automatically fill the required term with the proof of validity carried by the *yes* constructor. In case the transaction is not valid, we report a compile-time error.

We can now replace the operator for appending (valid) transactions to a ledger, with one that uses *implicit* tactic arguments instead:

```
_⊕_ : ValidLedger l
    → (tx : Tx)
    → {@(tactic validTxRefsM) : ∀ i → i ∈ inputs tx → Any (λ t → t ♯ ≡ id (outputRef i)) l
    → ...
    → ValidLedger (tx :: l)
(l ⊕ tx) {vtx} ... = l ⊕ tx ⊣ record { validTxRefs = vtx ; ... }
```

[11] https://github.com/agda/agda/blob/master/src/data/lib/prim/Agda/Builtin/Reflection.agda

SECTION 5

Formal Model II: BitML Calculus

Now let us shift our focus to our second subject of study, the BitML calculus for modelling smart contracts. In this subsection we sketch the formalized part of BitML we have covered so far, namely the syntax and small-step semantics of BitML contracts, as well as an example execution of a contract under these semantics. All code is publicly available on Github[12].

First, we begin with some basic definitions that will be used throughout this section:

```
module Types (Participant : Set) (Honest : List⁺ Participant) where

Time : Set
Time = ℕ

Value : Set
Value = ℕ

record Deposit : Set where
  constructor _has_
  field participant : Participant
        value : Value

Secret : Set
Secret = String

data Arith : List Secret → Set where ...
ℕ⟦_⟧ : ∀ {s} → Arith s → ℕ
ℕ⟦_⟧ = ...

data Predicate : List Secret → Set where ...
𝔹⟦_⟧ : ∀ {s} → Predicate s → Bool
𝔹⟦_⟧ = ...
```

[12] https://github.com/omelkonian/formal-bitml

Instead of giving a fixed datatype of participants, we parametrise our module with a given *universe* of participants and a non-empty list of honest participants. Representation of time and monetary values is again done using natural numbers, while we model participant secrets as simple strings[13]. A deposits consists of the participant that owns it and the number of bitcoins it carries. We, furthermore, introduce a simplistic language of logical predicates and arithmetic expressions with the usual constructs (e.g. numerical addition, logical conjunction) and give the usual semantics (predicates on booleans and arithmetic on naturals). A more unusual feature of these expressions is the ability to calculate length of secrets (within arithmetic expressions) and, in order to ensure more type safety later on, all expressions are indexed by the secrets they internally use.

5.1 Contracts in BitML

A *contract advertisement* consists of a set of *preconditions*, which require some resources from the involved participants prior to the contract's execution, and a *contract*, which specifies the rules according to which bitcoins are transferred between participants.

Preconditions either require participants to have a deposit of a certain value on their name (volatile or not) or commit to a certain secret. Notice the index of the datatype below, which captures the values of all required deposits:

```
data Precondition : List Value → Set where
  -- volatile deposit
  _?_ : Participant → (v : Value) → Precondition [ v ]
  -- persistent deposit
  _!_ : Participant → (v : Value) → Precondition [ v ]
  -- committed secret
  _#_ : Participant → Secret → Precondition [ ]
  -- conjunction
  _∧_ : Precondition vs₁ → Precondition vs₂ → Precondition (vs₁ ++ vs₂)
```

Moving on to actual contracts, we define them by means of a collection of five types of commands; *put* injects participant deposits and revealed secrets in the remaining contract, *withdraw* transfers the current funds to a participant, *split* distributes the current funds across different individual contracts, _:_ requires the authorization from a participant to proceed and *after* _:_ allows further execution of the contract only after some time has passed.

```
data Contract : Value  -- the monetary value it carries
              → Values -- the deposits it presumes
              → Set where
  -- collect deposits and secrets
  put _ reveal _ if _ ⇒ _ ⊣ _ :
```

[13] Of course, one could provide more realistic types (e.g. words of specific length) to be closer to the implementation, as shown for the UTxO model in Section **??**.

```
    (vs : List Value) → (s : Secrets) → Predicate s′ → Contract (v + sum vs) vs′ → s′ ⊆ s
    → Contract v (vs′ ++ vs)
  -- transfer the remaining balance to a participant
  withdraw : ∀ {v} → Participant → Contract v []
  -- split the balance across different branches
  split : (cs : List (∃[ v ] ∃[ vs ] Contract v vs))
    → Contract (sum (proj₁ ⟨$⟩ cs)) (concat (proj₂ ⟨$⟩ cs))
  -- wait for participant's authorization
  _ : _ : Participant → Contract v vs → Contract v vs
  -- wait until some time passes
  after _ : _ : Time → Contract v vs → Contract v vs
```

There is a lot of type-level manipulation across all constructors, since we need to make sure that indices are calculated properly. For instance, the total value in a contract constructed by the *split* command is the sum of the values carried by each branch. The *put* command[14] additionally requires an explicit proof that the predicate of the *if* part only uses secrets revealed by the same command.

We also introduce an intuitive syntax for declaring the different branches of a *split* command, emphasizing the *linear* nature of the contract's total monetary value:

```
_—∘_ : ∀ {vs : Values} → (v : Value) → Contract v vs → ∃[ v ] ∃[ vs ] Contract v vs
_—∘_ {vs} v c = v , vs , c
```

Having defined both preconditions and contracts, we arrive at the definition of a contract advertisement:

```
record Advertisement (v : Value) (vsᶜ vsᵍ : List Value) : Set where
  constructor _⟨ _ ⟩⊣ _
  field G     : Precondition vs
        C     : Contracts v vs
        valid : length vsᶜ ⩽ length vsᵍ
              × participantsᵍ G ++ participantsᶜ C ⊆ (participant ⟨$⟩ persistentDepositsᵖ G)
```

Notice that in order to construct an advertisement, one has to also provide proof of the contract's validity with respect to the given preconditions, namely that all deposit references in the contract are declared in the precondition and each involved participant is required to have a persistent deposit.

To clarify things so far, let us see a simple example of a contract advertisement:

```
open BitML (A ∣ B) [A]⁺
ex-ad : Advertisement 5 [200] (200 :: 100 :: [])
```

[14] *put* comprises of several components and we will omit those that do not contain any helpful information, e.g. write *put* _ ⇒ _ when there are no revealed secrets and the predicate trivially holds.

```
ex-ad = ⟨ B ! 200 ∧ A ! 100 ⟩
          split ( 2 ⊸ withdraw B
                ⊕ 2 ⊸ after 100 : withdraw A
                ⊕ 1 ⊸ put [200] ⇒ B : withdraw {201} A ⊣ ...
                )
        ⊣ ...
```

We first need to open our module with a fixed set of participants (in this case *A* and *B*). We then define an advertisement, whose type already says a lot about what is going on; it carries ₿ 5, presumes the existence of at least one deposit of ₿ 200, and requires two deposits of ₿ 200 and ₿ 100.

Looking at the precondition itself, we see that the required deposits will be provided by *B* and *A*, respectively. The contract first splits the bitcoins across three branches: the first one gives ₿ 2 to *B*, the second one gives ₿ 2 to *A* after some time period, while the third one retrieves *B*'s deposit of ₿ 200 and allows *B* to authorise the withdrawal of the remaining funds (currently ₿ 201) from *A*.

We have omitted the proofs that ascertain the well-formedness of the *put* command and the advertisement, as they are straightforward and do not provide any more intuition[15].

5.2 Small-step Semantics

BitML is a *process calculus*, which is geared specifically towards smart contracts. Contrary to most process calculi that provide primitive operators for inter-process communication via message-passing [**?**], the BitML calculus does not provide such built-in features.

It, instead, provides domain-specific synchronization mechanisms through its *small-step* reduction semantics. These essentially define a *labelled transition system* between *configurations*, where *action* labels are emitted on every transition and represent the required actions of the participants. This symbolic model consists of two layers; the bottom one transitioning between *untimed* configurations and the top one that works on *timed* configurations.

We start with the datatype of actions, which showcases the principal actions required to satisfy an advertisement's preconditions and an action to pick one branch of a collection of contracts (introduced by the choice operator ⊕). We have omitted uninteresting actions concerning the manipulation of deposits, such as dividing, joining, donating and destroying them. Since we will often need versions of the types of advertisements/contracts with their indices existentially quantified, we first provide aliases for them.

```
AdvertisedContracts : Set
AdvertisedContracts = List (∃[ v ] ∃[ vsᶜ ] ∃[ vsᵍ ] Advertisement v vsᶜ vsᵍ)

ActiveContracts : Set
ActiveContracts = List (∃[ v ] ∃[ vs ] List (Contract v vs))
```

[15] In fact, we have defined decidable procedures for all such proofs using the *proof-by-reflection* pattern [Van Der Walt and Swierstra 2012]. These automatically discharge all proof obligations, when there are no variables involved.

```
data Action (p : Participant) -- the participant that authorises this action
  : AdvertisedContracts       -- the contract advertisments it requires
  → ActiveContracts           -- the active contracts it requires
  → Values                    -- the deposits it requires from this participant
  → List Deposit              -- the deposits it produces
  → Set where

  -- commit secrets to stipulate an advertisement
  ♯▷ _ : (ad : Advertisement v vsᶜ vsᵍ)
       → Action p [v , vsᶜ , vsᵍ , ad] [] [] []
  -- spend x to stipulate an advertisement
  _ ▷ˢ _ : (ad : Advertisement v vsᶜ vsᵍ)
         → (i : Index vsᵍ)
         → Action p [v , vsᶜ , vsᵍ , ad] [] [vsᵍ !! i] []
  -- pick a branch
  _ ▷ᵇ _ : (c : List (Contract v vs))
         → (i : Index c)
         → Action p [] [v , vs , c] [] []
  ⋮
```

The action datatype is parametrised[16] over the participant who performs it and includes several indices representing the prerequisites the current configuration has to satisfy, in order for the action to be considered valid (e.g. one cannot spend a deposit to stipulate an advertisement that does not exist).

The first index refers to advertisements that appear in the current configuration, the second to contracts that have already been stipulated, the third to deposits owned by the participant currently performing the action and the fourth declares new deposits that will be created by the action (e.g. dividing a deposit would require a single deposit as the third index and produce two other deposits in its fourth index).

Although our indexing scheme might seem a bit heavyweight now, it makes many little details and assumptions explicit, which would bite us later on when we will need to reason about them.

Continuing from our previous example advertisement, let's see an example action where A spends the required ₿ 100 to stipulate the example contract[17]:

```
ex-spend : Action A [5 , [200] , 200 :: 100 :: [] , ex-ad] [] [100] []
ex-spend = ex-ad ▷ˢ 1
```

[16] In Agda, datatype parameters are similar to indices, but are not allowed to vary across constructors.

[17] Notice that we have to make all indices of the advertisement explicit in the second index in the action's type signature.

Configurations are now built from advertisements, active contracts, deposits, action authorizations and committed/revealed secrets:

```
data Configuration′ : --          current          ×      required
                      AdvertisedContracts × AdvertisedContracts
                    → ActiveContracts      × ActiveContracts
                    → List Deposit         × List Deposit
                    → Set where

  -- empty
  ∅ : Configuration′ ([] , []) ([] , []) ([] , [])

  -- contract advertisement
  ‘_ : (ad : Advertisement v vsᶜ vsᵍ)
       → Configuration′ ([ v , vsᶜ , vsᵍ , ad ] , []) ([] , []) ([] , [])

  -- active contract
  ⟨ _ , _ ⟩ᶜ : (c : List (Contract v vs)) → Value
               → Configuration′ ([] , []) ([ v , vs , c ] , []) ([] , [])

  -- deposit redeemable by a participant
  ⟨ _ , _ ⟩ᵈ : (p : Participant) → (v : Value)
               → Configuration′ ([] , []) ([] , []) ([ p has v ] , [])

  -- authorization to perform an action
  _[_]  : (p : Participant) → Action p ads cs vs ds
          → Configuration′ ([] , ads) ([] , cs) (ds , ((p has _)⟨$⟩vs))

  -- committed secret
  ⟨ _ :_ ♯_ ⟩ : Participant → Secret →      ℕ ⊎ ⊥
                → Configuration′ ([] , []) ([] , []) ([] , [])

  -- revealed secret
  _ :_ ♯_ : Participant → Secret → ℕ
            → Configuration′ ([] , []) ([] , []) ([] , [])

  -- parallel composition
  _ | _ : Configuration′ (adsˡ , radsˡ) (csˡ , rcsˡ) (dsˡ , rdsˡ)
          → Configuration′ (adsʳ , radsʳ) (csʳ , rcsʳ) (dsʳ , rdsʳ)
          → Configuration′ (adsˡ          ++ adsʳ , radsˡ ++ (radsʳ \ adsˡ))
                           (csˡ           ++ csʳ  , rcsˡ  ++ (rcsʳ  \ csˡ))
                           ((dsˡ \ rdsʳ) ++ dsʳ   , rdsˡ  ++ (rdsʳ  \ dsˡ))
```

The indices are quite involved, since we need to record both the current advertisements, stipulated contracts and deposits and the required ones for the configuration to become valid. The most interesting case is the parallel composition operator, where the resources provided by the left operand might satisfy some requirements of the right operand. Moreover, consumed deposits have

to be eliminated as there can be no double spending, while the number of advertisements and contracts always grows.

By composing configurations together, we will eventually end up in a *closed* configuration, where all required indices are empty (i.e. the configuration is self-contained):

```
Configuration : AdvertisedContracts → ActiveContracts → List Deposit → Set
Configuration ads cs ds = Configuration′ (ads , []) (cs , []) (ds , [])
```

We are now ready to declare the inference rules of the bottom layer of our small-step semantics, by defining an inductive datatype modelling the binary step relation between untimed configurations:

```
data _ ⟶ _ : Configuration ads cs ds → Configuration ads′ cs′ ds′ → Set where
  DEP-AuthJoin :
    ⟨ A , v ⟩ᵈ | ⟨ A , v′ ⟩ᵈ | Γ ⟶ ⟨ A , v ⟩ᵈ | ⟨ A , v′ ⟩ᵈ | A [0 ↔ 1] | Γ

  DEP-Join :
    ⟨ A , v ⟩ᵈ | ⟨ A , v′ ⟩ᵈ | A [0 ↔ 1] | Γ ⟶ ⟨ A , v + v′ ⟩ᵈ | Γ

  C-Advertise : ∀ {Γ ad}
    → ∃[p ∈ participantsᵍ (G ad)] p ∈ Hon
    ──────────────────────────────────────
    → Γ ⟶ ‘ad | Γ

  C-AuthCommit : ∀ {A ad Γ}
    → secrets (G ad) ≡ a₀ ... aₙ
    → (A ∈ Hon → ∀ [i ∈ 0 ... n] aᵢ ≢ ⊥)
    ──────────────────────────────────────
    → ‘ad | Γ ⟶ ‘ad | Γ | ... ⟨ A : aᵢ ♯Nᵢ ⟩ ... | A [♯▷ ad]

  C-Control : ∀ {Γ C i D}
    → C !! i ≡ A₁ : A₂ : ... : A : D
    ──────────────────────────────────────
    → ⟨ C , v ⟩ᶜ | ... Aᵢ [C ▷ᵇ i] ... | Γ ⟶ ⟨ D , v ⟩ᶜ | Γ
  ⋮
```

There is a total of 18 rules we need to define, but we choose to depict only a representative subset of them. The first pair of rules initially appends the authorisation to merge two deposits to the current configuration (rule *DEP-AuthJoin*) and then performs the actual join (rule [*DEP-Join*]). This

is a common pattern across all rules, where we first collect authorisations for an action by all involved participants, and then we fire a subsequent rule to perform this action. [C-Advertise] advertises a new contract, mandating that at least one of the participants involved in the pre-condition is honest and requiring that all deposits needed for stipulation are available in the surrounding context. [C-AuthCommit] allows participants to commit to the secrets required by the contract's pre-condition, but only dishonest ones can commit to the invalid length ⊥. Lastly, [C-Control] allows participants to give their authorization required by a particular branch out of the current choices present in the contract, discarding any time constraints along the way.

It is noteworthy to mention that during the transcriptions of the complete set of rules from the paper [?] to our dependently-typed setting, we discovered a discrepancy in the [C-AuthRev] rule, namely that there was no context Γ. Moreover, in order to later facilitate equational reasoning, we re-factored the [C-Control] to not contain the inner step as a hypothesis, but instead immediately inject it in the result operand of the step relation.

The inference rules above have elided any treatment of timely constraints; this is handled by the top layer, whose states are now timed configurations. The only interesting inference rule is the one that handles time decorations of the form *after* _ : _, since all other cases are dispatched to the bottom layer (which just ignores timely aspects).

```
record Configurationᵗ (ads : AdvertisedContracts) (cs : ActiveContracts) (ds : Deposits) : Set where
  constructor _ @ _
  field cfg  : Configuration ads cs ds
        time : Time

data _ ⟶ₜ _ : Configurationᵗ ads cs ds → Configurationᵗ ads′ cs′ ds′ → Set where
  Action : ∀ {Γ Γ′ t}
    → Γ ⟶ Γ′
    ──────────────────
    → Γ @ t ⟶ₜ Γ′ @ t
  Delay : ∀ {Γ t δ}
    ──────────────────────────
    → Γ @ t ⟶ₜ Γ @ (t + δ)
  Timeout : ∀ {Γ Γ′ t i contract}
    → All (_ ⩽ t) (timeDecorations (contract !! i)) -- all time constraints are satisfied
    → ⟨ [ contract !! i ] , v ⟩ᶜ | Γ ⟶ Γ′  -- resulting state if we pick this branch
    ──────────────────────────
    → (⟨ contract , v ⟩ᶜ | Γ) @ t ⟶ₜ Γ′ @ t
```

Having defined the step relation in this way allows for equational reasoning, a powerful tool for writing complex proofs:

```
data _ ↠ _ : Configuration ads cs ds → Configuration ads′ cs′ ds′ → Set where
  _□ : (M : Configuration ads cs ds) → M ↠ M
  _ ⟶ ⟨ _ ⟩_ : ∀ {M N} (L : Configuration ads cs ds)
    → L ⟶ M → M ↠ N
      ──────────────
    → L ↠ N
begin_ : ∀ {M N} → M ↠ N → M ↠ N
```

5.3 Example

We are finally ready to see a more intuitive example of the *timed-commitment protocol*, where a participant commits to revealing a valid secret a (e.g. "qwerty") to another participant, but loses her deposit of ₿ 1 if she does not meet a certain deadline t:

```
tc : Advertisement 1 [] (1 :: 0 :: [])
tc = ⟨ A ! 1 ∧ A ♯ a ∧ B ! 0 ⟩ reveal [ a ] ⇒ withdraw A ⊣ ... ⊕ after t : withdraw B
```

Below is one possible reduction in the bottom layer of our small-step semantics, demonstrating the case where the participant actually meets the deadline:

```
tc-semantics : ⟨ A , 1 ⟩ᵈ ↠ ⟨ A , 1 ⟩ᵈ | A : a ♯6
tc-semantics =
  begin
    ⟨ A , 1 ⟩ᵈ
  ⟶⟨ C-Advertise ⟩
    ‘tc | ⟨ A , 1 ⟩ᵈ
  ⟶⟨ C-AuthCommit ⟩
    ‘tc | ⟨ A , 1 ⟩ᵈ | ⟨ A : a ♯6 ⟩ | A [ ♯ ▷ tc ]
  ⟶⟨ C-AuthInit ⟩
    ‘tc | ⟨ A , 1 ⟩ᵈ | ⟨ A : a ♯6 ⟩ | A [ ♯ ▷ tc ] | A [ tc ▷ˢ 0 ]
  ⟶⟨ C-Init ⟩
    ⟨ tc , 1 ⟩ᶜ | ⟨ A : a ♯ inj₁ 6 ⟩
  ⟶⟨ C-AuthRev ⟩
    ⟨ tc , 1 ⟩ᶜ | A : a ♯6
  ⟶⟨ C-Control ⟩
    ⟨ [ reveal [ a ] ⇒ withdraw A ⊣ ... ] , 1 ⟩ᶜ | A : a ♯6
  ⟶⟨ C-PutRev ⟩
    ⟨ [ withdraw A ] , 1 ⟩ᶜ | A : a ♯6
  ⟶⟨ C-Withdraw ⟩
```

$$\langle\, A \,,\, 1 \,\rangle^{\mathrm{d}} \mid A : a \sharp 6$$

□

At first, A holds a deposit of ₿ 1, as required by the contract's precondition. Then, the contract is advertised and the participants slowly provide the corresponding prerequisites (i.e. A commits to a secret via *C-AuthCommit* and spends the required deposit via *C-AuthInit*, while B does not do anything). After all pre-conditions have been satisfied, the contract is stipulated (rule *C-Init*) and the secret is successfully revealed (rule *C-AuthRev*). Finally, the first branch is picked (rule *C-Control*) and A retrieves her deposit back (rules *C-PutRev* and *C-Withdraw*).

5.4 Reasoning Modulo Permutation

In the definitions above, we have assumed that $(\_|\_\,,\,\varnothing)$ forms a commutative monoid, which allowed us to always present the required sub-configuration individually on the far left of a composite configuration. While such definitions enjoy a striking similarity to the ones appearing in the original paper [**?**] (and should always be preferred in an informal textual setting), this approach does not suffice for a mechanized account of the model. After all, this explicit treatment of all intuitive assumptions/details is what makes our approach robust and will lead to a deeper understanding of how these systems behave. To overcome this intricacy, we introduce an *equivalence relation* on configurations, which holds when they are just permutations of one another:

```
_≈_ : Configuration ads cs ds → Configuration ads cs ds → Set
c ≈ c′ = cfgToList c ↭ cfgToList c′
  where
    open import Data.List.Permutation using (_↭_)

    cfgToList : Configuration′ p₁ p₂ p₃ → List (∃[p₁] ∃[p₂] ∃[p₃] Configuration′ p₁ p₂ p₃)
    cfgToList ∅           = []
    cfgToList (l ∣ r)     = cfgToList l ++ cfgToList r
    cfgToList {p₁} {p₂} {p₃} c = [ p₁ , p₂ , p₃ , c ]
```

Given this reordering mechanism, we now need to generalise all our inference rules to implicitly reorder the current and next configuration of the step relation. We achieve this by introducing a new variable for each of the operands of the resulting step relations, replacing the operands with these variables and requiring that they are re-orderings of the previous configurations, as shown in the following generalisation of the *DEP-AuthJoin* rule[18]:

$$\frac{\begin{array}{l}\Gamma' \approx \langle\, A \,,\, v \,\rangle^{\mathrm{d}} \mid \langle\, A \,,\, v' \,\rangle^{\mathrm{d}} \mid \Gamma \quad \in \mathit{Configuration}\ \mathit{ads}\ \mathit{cs}\ (A\ \mathit{has}\ v :: A\ \mathit{has}\ v' :: \mathit{ds}) \\ \to \Gamma'' \approx \langle\, A \,,\, v \,\rangle^{\mathrm{d}} \mid \langle\, A \,,\, v' \,\rangle^{\mathrm{d}} \mid A\ [0 \leftrightarrow 1] \mid \Gamma \ \in \mathit{Configuration}\ \mathit{ads}\ \mathit{cs}\ (A\ \mathit{has}\ (v + v')\ :: \mathit{ds})\end{array}}{\to \Gamma' \longrightarrow \Gamma''}$$

[18] In fact, it is not necessary to reorder both ends for the step relation; at least one would be adequate.

Unfortunately, we now have more proof obligations of the re-ordering relation lying around, which makes reasoning about our semantics rather tedious. We are currently investigating different techniques to model such reasoning up to equivalence:

- *Quotient types* [?] allow equipping a type with an equivalence relation. If we assume the axiom that two elements of the underlying type are *propositionally* equal when they are equivalent, we could discharge our current proof burden trivially by reflexivity. Unfortunately, while one can easily define *setoids* in Agda, there is not enough support from the underlying type system to make reasoning about such an equivalence as easy as with built-in equality.
- Going a step further into more advanced notions of equality, we arrive at *homotopy type theory* [?], which tries to bridge the gap between reasoning about isomorphic objects in informal pen-paper proofs and the way we achieve this in mechanized formal methods. Again, realizing practical systems with such an enriched theory is a topic of current research [?] and no mature implementation exists yet, so we cannot integrate it with our current development in any pragmatic way.
- The crucial problems we have encountered so far are attributed to the non-deterministic nature of BitML, which is actually inherent in any process calculus. Building upon this idea, we plan to take a step back and investigate different reasoning techniques for a minimal process calculus. Once we have an approach that is more suitable, we will incorporate it in our full-blown BitML calculus.

5.5 Symbolic Model

In order to formalize the BitML's symbolic model, we first notice that a constructed derivation witnesses one of many possible contract executions. In other words, derivations of our small-step semantics model *traces* of the contract execution. Our symbolic model will provide a game-theoretic view over those traces, where each participant has a certain *strategy* that selects moves depending on the current trace of previous moves. Moves here should be understood just as emissions of a label, i.e. application of a certain inference rule.

5.5.1 Labelled Step Relation

To that end, we associate a label to each inference rule and extend the original step relation to additionally emit labels, hence defining a *labelled transition system*.

We first define the set of labels, which basically distinguish which rule was used, along with all (non-proof) arguments that are required by the rule:

```
data Label : Set where

  auth-join [ _ , _ ↔ _ ] : Participant → DepositIndex → DepositIndex → Label
  join [ _ ↔ _ ] :                        DepositIndex → DepositIndex → Label

  auth-divide [ _ , _▷_ , _ ] : Participant → DepositIndex → Value → Value → Label
  divide [ _▷_ , _ ] :                      DepositIndex → Value → Value → Label
```

```
auth-donate [ _ , _ ▷ᵈ _ ] : Participant → DepositIndex → Participant → Label
donate [ _ ▷ᵈ _ ] :                      DepositIndex → Participant → Label

auth-destroy [ _ , _ ] : Participant → DepositIndex → Label
destroy [ _ ] : DepositIndex → Label

advertise [ _ ] : ∃Advertisement → Label

auth-commit [ _ , _ , _ ] : Participant → ∃Advertisement → List CommittedSecret → Label
auth-init [ _ , _ , _ ] : Participant → ∃Advertisement → DepositIndex → Label
init [ _ ] : ∃Advertisement → Label

split : Label

auth-rev [ _ , _ ] : Participant → Secret → Label
rev [ _ , _ ] : Values → Secrets → Label

withdraw [ _ , _ ] : Participant → Value → Label

auth-control [ _ , _ ▷ᵇ _ ] : Participant → (c : ∃Contracts) → Index (proj₂ (proj₂ c)) → Label
control : Label

delay [ _ ] : Time → Label
```

Notice how we existentially pack indexed types, so that *Label* remains simply-typed. This is essential, as it would be tedious to manipulate indices when there is no need for them. Moreover, some indices are now just $\mathbb{N}$ instead of *Fin*, losing the guarantee to not fall out-of-bounds.

The step relation will now emit the corresponding label for each rule. Below, we give the updated kind signature and an example for the *DEP-AuthJoin* rule:

```
data _ ⟶⟦ _ ⟧ _ : Configuration ads cs ds
                → Label
                → Configuration ads′ cs′ ds′
                → Set where
  ⋮
  DEP-AuthJoin :
    ⟨ A , v ⟩ᵈ | ⟨ A , v′ ⟩ᵈ | Γ
  ⟶⟦ auth-join [A , 0 ↔ 1] ⟧
    ⟨ A , v ⟩ᵈ | ⟨ A , v′ ⟩ᵈ | A [0 ↔ 1] | Γ
  ⋮
```

Naturally, the reflexive transitive closure of the augmented step relation will now hold a sequence of labels as well:

```
data _ ↠⟦ _ ⟧ _ : Configuration ads cs ds
                → Labels
                → Configuration ads′ cs′ ds′
                → Set where

  _□ : (M : Configuration ads cs ds)
       ──────────────────────────────
     → M ↠⟦ [] ⟧ M

  _ ⟶⟨ _ ⟩⊢ _ : (L : Configuration ads cs ds) {L′ : Configuration ads cs ds}
                 {M M′ : Configuration ads′ cs′ ds′ } {N : Configuration ads″ cs″ ds″ }
     → L′ ⟶⟦ a ⟧ M′
     → (L ≈ L′ ) × (M ≈ M′ )
     → M ↠⟦ as ⟧ N
       ──────────────────────────────────────────
     → L ↠⟦ a :: as ⟧ N

start : {M : Configuration ads cs ds} {N : Configuration ads′ cs′ ds′ }
  → M ↠⟦ as ⟧ N
    ──────────
  → M ↠⟦ as ⟧ N
start M ↠ N = M ↠ N
```

The timed variants of the step relation follow exactly the same procedure, so we do not repeat the definitions here.

5.5.2 Traces

Values of type _ ↠⟦ _ ⟧ _ model execution traces. Since the complex type indices of the step-relation datatype is not as useful here, we define a simpler datatype of execution traces that is a list of labelled transitions between (existentially-packed) timed configurations:

```
data Trace : Set where
  _•          : ∃TimedConfiguration → Trace

  _ ::⟦ _ ⟧ _ : ∃TimedConfiguration → Label → Trace → Trace
```

Stripping. Strategies will make moves based on these traces, so we need a *stripping* operation that traverses a configuration with its emitted labels and removes any sensitive information (i.e. committed secrets):

```
stripCfg : Configuration′ p₁ p₂ p₃ → Configuration′ p₁ p₂ p₃
stripCfg⟨ p : a ♯ _ ⟩ = ⟨ p : a ♯ nothing ⟩
stripCfg (l | r ⊣ p) = stripCfg l | stripCfg r ⊣ p
stripCfg c           = c

stripLabel : Label → Label
stripLabel auth-commit [ p , ad , _ ] = auth-commit [ p , ad , [] ]
stripLabel a = a

_ ∗ :Trace → Trace
(... , Γ @ t) ∗          = (... , stripCfg Γ @ t)
(... , Γ @ t) ::⟦ α ⟧ ts = (... , stripCfg Γ @ t) ::⟦ stripLabel α ⟧ (ts ∗)
```

5.5.3 Strategies

Participant strategies are functions which, given the (stripped) trace so far, pick a set of possible next moves for its participant. These moves cannot be arbitrary; they have to satisfy several validity conditions which we require as proof in the datatype definition itself.

Strategies are expected to be PPTIME algorithms, so as to have a certain computational bound on the processing they can undergo to compute secrets, etc. Since working on a resource-aware logic would make this much more difficult in search of tooling and infrastructure, we ignore this requirement and simply model strategies as regular functions.

Before we define the types of strategies, we give a convenient notation to extend a trace with another (timed) transition:

```
_ ↣⟦ _ ⟧ _ : Trace → Label → ∃TimedConfiguration → Set
R ↣⟦ α ⟧ (_ , _ , _ , tc′ )
  = proj₂ (proj₂ (proj₂ (lastCfg R))) ⟶⟦ α ⟧ tc′
```

Honest strategies. Each honest participant is modelled by a symbolic strategy that outputs a set of possible next moves with respect to the current trace. These moves have to be *valid*, thus we define *honest strategies* as a dependent record:

```
record HonestStrategy (A : Participant) : Set where
  field
    strategy : Trace → Labels
    valid    : A ∈ Hon                                        (1)
             × (∀ R α → α ∈ strategy (R ∗) →                  (2)
                 ∃[ R′ ] (R ↣⟦ α ⟧ R′ ))
```

```
  × (∀ R α → α ∈ strategy (R ∗) →                                    (3)
       All (_ ≡ A) (authDecoration α))
  × (∀ R Δ Δ′ ad →                                                  (4)
       auth-commit [A , ad , Δ] ∈ strategy (R ∗) →
       auth-commit [A , ad , Δ′ ] ∈ strategy (R ∗) →
         Δ ≡ Δ′ )
  × (∀ R T′ α → α ∈ strategy (R ∗) →                                 (5)
       ∃[α′ ] (R ↣⟦ α′ ⟧ T′ ) →
       ∃[R″] (T′ ::⟦ α ⟧ R ↣⟦ α ⟧ R″) →
         α ∈ strategy ((T′ ::⟦ α ⟧ R) ∗))
```

Condition (1) restricts our participants to the honest subset[19] and condition (2) requires that chosen moves are in accordance to the small-step semantics of BitML. Condition (3) states that one cannot authorize moves for other participants, condition (4) requires that the lengths of committed secrets are *coherent* (i.e. no different lengths for the same secrets across moves) and condition (5) dictates that decisions are *consistent*, so as moves that are not chosen will still be selected by the strategy in a future run (if they are still valid).

All honest participants should be accompanied by such a strategy, so we pack all honest strategies in one single datatype:

```
HonestStrategies : Set
HonestStrategies = ∀ {A} → A ∈ Hon → ParticipantStrategy A
```

Adversary strategies. All dishonest participant will be modelled by a single adversary *Adv*, whose strategy now additionally takes the moves chosen by the honest participants and makes the final decision.

Naturally, the chosen move is subject to certain conditions and is again a dependent record:

```
record AdversarialStrategy (Adv : Participant) : Set where
  field
    strategy : Trace → List (Participant × Labels) → Label
    valid    : Adv ∉ Hon                                              (1)
             × (∀ {B ad Δ} → B ∉ Hon → α ≡ auth-commit [B , ad , Δ] →  (2)
                 α ≡ strategy (R ∗) [])
             × ∀ {R : Trace} {moves : List (Participant × Labels)} →  (3)
                 let α = strategy (R ∗) moves in
                 ( ∃[A]
                     ( A ∈ Hon
                     × authDecoration α ≡ just A
```

[19] Recall that *Hon* is non-empty, i.e. there is always at least one honest participant.

```
        × α ∈ concatMap proj₂ moves)
⊎ (     authDecoration α ≡ nothing
  × (∀ δ → α ≢ delay [ δ ])
  × ∃[ R′ ] (R ⟼⟦ α ⟧ R′ ))
⊎ (∃[ B ]
      (     (authDecoration α ≡ just B)
        × (B ∉ Hon)
        × (∀ s → α ≢ auth-rev [ B , s ])
        × ∃[ R′ ] (R ⟼⟦ α ⟧ R′ )))
⊎ ∃[ δ ]
      (  (α ≡ delay [ δ ])
      × All (λ { (_ , ) → ( ≡ [ ]) ⊎ Any (λ { delay [ δ′ ] → δ′ ≥δ; _ → ⊥ }) }) moves)
⊎ ∃[ B ]∃[ s ]
      (  α ≡ auth-rev [ B , s ]
      × B ∉ Hon
      × ⟨ B : s ♯ nothing ⟩ ∈ (R ∗)
      × ∃[ R∗′ ]∃[ Δ ]∃[ ad ]
            (  R∗′ ∈ prefixTraces (R ∗)
            × strategy R∗′ [ ] ≡ auth-commit [ B , ad , Δ ]
            × (s , nothing) ∈ Δ)))
```

The first two conditions state that the adversary is not one of the honest participants and that committing cannot depend on the honest moves, respectively. Condition (3) constraints the move that is chosen by the adversary, such that one of the following conditions hold:

(1) The move was chosen out of the available honest moves.
(2) It is not a *delay*, nor does it require any authorization.
(3) It is authorized by a dishonest participant, but is not a secret-revealing move.
(4) It is a *delay*, but one that does not influence the time constraints of the honest participants.
(5) It reveals a secret from a dishonest participant, in which case there is valid commit (i.e. with non-$\bot$ length) somewhere in the previous trace.

A complete set of strategies includes a strategy for each honest participant and a single adversarial strategy:

```
Strategies : Set
Strategies = AdversarialStrategy × HonestStrategies
```

We can now describe how to proceed execution on the current trace, namely by retrieving possible moves from all honest participants and giving control to the adversary to make the final choice for a label:

```
runAdversary : Strategies → Trace → Label
runAdversary (S† , S) R = strategy S†(R ∗) (runHonestAll (R ∗) S)
  where
    runHonestAll : Trace → List (Participant × Labels) → HonestMoves
    runHonestAll R S = mapWith ∈ Hon (λ {A} A ∈→ A , strategy (S A ∈) (R ∗))
```

Symbolic Conformance. Given a trace, we can formulate a notion of *conformance* of a trace with respect to a set of strategies, namely when we transitioned from an initial configuration to the current trace using only moves obtained by those strategies:

```
data _-conforms-to-_ : Trace → Strategies → Set where

  base : ∀ {Γ : Configuration ads cs ds} {SS : Strategies}
    → Initial Γ
    ---------------------------------------
    → (ads , cs , ds , Γ @ 0) •  -conforms-to-  SS

  step : ∀ {R : Trace} {T′ : ∃TimedConfiguration} {SS : Strategies}
    → R -conforms-to-SS
    → R ↣⟦ runAdversary SS R ⟧ T′
    ---------------------------------------
    → (T′ ::⟦ runAdversary SS R ⟧ R)  -conforms-to-  SS
```

5.5.4 *Meta-theoretical results*

To increase confidence in our symbolic model, we proceed with the mechanization of two meta-theoretical lemmas.

Stripping preserves semantics. The first one concerns the operation of stripping sensitive values out of a trace. If we exclude moves that reveal or commit secrets (i.e. rules *AuthRefv* and *AuthCommit*), we can formally prove that stripping preserves the small-step semantics:

```
∗ − preserves-semantics :
  (∀ A s      → α ≢ auth-rev [A , s]) →
  (∀ A ad Δ → α ≢ auth-commit [A , ad , Δ])
  → (∀ T′  → R ↣⟦ α ⟧ T′
           ---------------------
           → R ∗ ↣⟦ α ⟧ T′ ∗)
  × (∀ T′  → R ∗ ↣⟦ α ⟧ T′
```

$$\rightarrow \exists[\, T'' \,]\ (R \rightarrowtail\!\!\!\!\!\!\rightarrow[\![\ \alpha\ ]\!]\ T'')\ \times\ (T'\ *\ \equiv\ T''\ *)$$

The second part of the conclusion states that if we have a transition from a stripped state, then there is an equivalent target state (modulo additional sensitive information) to which the un-stripped state can transition.

Adversarial moves are always semantic. Lastly, it holds that all moves that can be chosen by the adversary are admitted by the small-step semantics:

$$\begin{array}{l} \mathit{adversarial\text{-}move\text{-}is\text{-}semantic} : \\ \quad \exists[\, T' \,]\ (R \rightarrowtail\!\!\!\!\!\!\rightarrow[\![\ \mathit{runAdversary}\ (S\dagger\ ,\ S)\ R\ ]\!]\ T'\ ) \end{array}$$

5.6 BitML Paper Fixes

It is expected in any mechanization of a substantial amount of theoretical work to encounter inconsistencies in the pen-and-paper version, ranging from simple typos and omissions to fundamental design problems. This is certainly one of the primary selling points for formal verification; corner cases that are difficult to find by testing or similar methods, can instead be discovered with rigorous formal methods.

Our formal development was no exception, since we encountered several issues with the original presentation, which led to the modifications presented below.

Inference Rules. Rule *DEP-Join* requires two symmetric invocations of the *DEP-AuthJoin* rule, but it is unclear if this gives us anything meaningful. Instead, we choose to simplify the rule by requiring just one authorization.

When rule *C-AuthRev* is presented in the original BitML paper, it seems to act on an atomic configuration $\langle\ A : \alpha \sharp \mathbb{N}\ \rangle$. This renders the rule useless in any practical scenario, so we extend the rule to include a surrounding context:

$$\langle\ A : s \sharp \mathit{just}\ n\ \rangle \| \Gamma \longrightarrow[\![\ \mathit{auth\text{-}rev}\ [A\ ,\ s]\ ]\!]\ A : s \sharp n \| \Gamma$$

Small-step Derivations as Equational Reasoning. In Section **??** , we saw an example derivation of our small-step semantics, given in an equational-reasoning style. This is possible, because the involved rules follow a certain format.

Alas, rule *C-Control* includes another transition in its premises which results in the same state Γ' as the transition in the conclusion, resulting in a tree-like proof structure. which is arguably inconvenient for textual presentation. This is problematic when we try to reason in an equational-reasoning style using our multi-step relation $\_ \twoheadrightarrow \_$, since this branching will break our sequential way of presenting the proof step by step.

To avoid this issue, notice how we can "linearize" the proof structure by removing the premise and replacing the target configuration of the conclusion with the source configuration of the removed premise. Our version of *C-Control* reflects this important refactoring.

Conditions for Adversarial Strategies. Moves chosen by an adversarial strategy come in two forms: labels and pairs (A, j) of an honest participant A with an index into his/her current moves. However, this is unnecessary, since we can both cases uniformly using our *Label* type.

Semantics-preserving Stripping. The meta-theoretical lemma concerning stripping in the original paper (*Lemma 3*) requires that the transition considered is not an application of the *Auth-Rev* rule. It turns out this is not a strong enough guarantee, since the *AuthCommit* rule also contains sensitive information, thus would not be preserved after stripping. We, therefore, fix the statement in Lemma 3 to additionally require that $\alpha \not\equiv A : \langle\, G \,\rangle C, \Delta$.

SECTION 6

Related Work

6.1 Static Analysis Tools

a TODO: BitML Liquidity ... TODO: Madmax ... TODO: Mooly ...

6.2 Scilla

TODO: extrinsic ...

6.3 Setzer's Agda model

TODO: similar, but ...

SECTION 7

Next steps

In this section, I describe possible next steps I plan to investigate during the remainder of my thesis. It is impossible to accurately predict what will be achieved in the following five months and there will definitely be some surprises along the way, but I hope it will give realistic expectations of the final results of my thesis.

7.1 Extended UTxO

7.1.1 2-level Multi-currency

TODO: ... non-fungible tokens ...

7.1.2 Multi-signature Scheme

TODO: ... where multiple parties have to synchonize authorisation ...

7.1.3 Plutus Integration

In my current formalization of the extended UTxO model, scripts are immediately modelled by their denotations (i.e. pure mathematical functions). This is not accurate, however, since scripts are actually pieces of program text. However, there is current development by James Chapman of IOHK to formalize the meta-theory of Plutus, Cardano's scripting language[20].

Since we mostly care about Plutus as a scripting language, it would be possible to replace the denotations with actual Plutus Core source code and utilize the formalized meta-theory to acquire the denotational semantics when needed.

7.2 BitML

7.2.1 Decision Procedures

TODO: more ergonomic proof-development process ...

7.2.2 Towards Completeness

Continuing my work on the formalization of the BitML paper [?], there is still a lot of theoretical results to be covered:

- While I currently have the symbolic model in place, there is still no formalization of *symbolic strategies*, where one can reason about different adversary strategies and prove that certain scenarios are impossible.

[20] https://github.com/input-output-hk/plutus-metatheory

- Another import task is to define the computational model; a counterpart of the symbolic model augmented with pragmatic computational properties to more closely resemble the low-level details of Bitcoin.
- When both symbolic and computational strategies have been formalized, I will be able to finally prove the correctness of the BitML compiler, which translates high-level BitML contracts to low-level standard Bitcoin transactions. The symbolic model concerns the input of the compiler, while the computational one concerns the output. This endeavour will involve implementing the actual translation and proving *coherence* between the symbolic and the computational model. Proving coherence essentially requires providing a (weak) *simulation* between the two models; each step in the symbolic part is matched by (multiple) steps in the computational one.

7.3 UTxO-BitML Integration

So far I have worked separately on the two models under study, but it would be interesting to see whether these can be intertwined in some way. This would possibly involve a translation from BitML contracts to contracts modelled in our extended UTxO models, along with corresponding meta-theoretical properties (e.g. validity of UTxO transactions correspond to another notion of validity of BitML contracts).

Moreover, and it would be beneficial to review the different modelling techniques used across both models, identifying their key strengths and witnesses. With this in mind, I could refactor crucial parts of each model for the sake of elegance, clarity and ease of reasoning.

TODO: BitML → eUTxO compiler … Compilation correctness … full abstraction

7.4 Featherweight Solidity

One of the posed research questions concerns the expressiveness of the extended UTxO model with respect to Ethereum-like account-based ledgers.

In order to investigate this in a formal manner, one has to initially model a reasonable subset of Solidity, so a next step would be to model *Featherweight Solidity*, taking inspiration from the approach taken in the formalization of Java using *Featherweight Java* [**?**]. Fortunately, I will not have to start from scratch, since there have been recent endeavours in F^* to analyse and verify Ethereum smart contracts, which already describe a simplified model of Solidity [**?**].

As a next step, one could try out different example contracts in Solidity and check whether they can be transcribed to contracts appropriate for an extended UTxO ledger.

7.5 Proof Automation

Last but not least, our current dependently-typed approach to formalizing our models has led to a significant proof burden, as evidenced by the complicated type signatures presented throughout this proposal. This certainly makes the reasoning process quite tedious and time consuming, so a reasonable task would be to implement automatic proof-search procedures using Agda meta-programming [**?**].

SECTION 8

Conclusion

References

2013. Homotopy Type Theory: Univalent Foundations of Mathematics. *CoRR* abs/1308.0729 (2013). arXiv:1308.0729 http://arxiv.org/abs/1308.0729

2019. The Extended UTxO Model. Retrieved 2/2019 from https://github.com/input-output-hk/plutus/blob/master/docs/extended-utxo/README.md

2019. Multi-Currency. Retrieved 5/2019 from https://github.com/input-output-hk/plutus/blob/master/docs/multi-currency/multi-currency.md

Thorsten Altenkirch, Thomas Anberrée, and Nuo Li. 2011. Definable quotients in type theory. *Draft paper* (2011), 48–49.

Marcin Andrychowicz, Stefan Dziembowski, Daniel Malinowski, and Lukasz Mazurek. 2014. Secure multiparty computations on bitcoin. In *Security and Privacy (SP), 2014 IEEE Symposium on*. IEEE, 443–458.

Bruno Barras, Samuel Boutin, Cristina Cornes, Judicaël Courant, Jean-Christophe Filliatre, Eduardo Gimenez, Hugo Herbelin, Gerard Huet, Cesar Munoz, Chetan Murthy, et al. 1997. *The Coq proof assistant reference manual: Version 6.1*. Ph.D. Dissertation. Inria.

Massimo Bartoletti and Roberto Zunino. 2018. *BitML: a calculus for Bitcoin smart contracts*. Technical Report. Cryptology ePrint Archive, Report 2018/122.

Iddo Bentov and Ranjit Kumaresan. 2014. How to use bitcoin to design fair protocols. In *International Cryptology Conference*. Springer, 421–439.

Karthikeyan Bhargavan, Antoine Delignat-Lavaud, Cedric Fournet, A Gollamudi, G Gonthier, N Kobeissi, A Rastogi, T Sibut-Pinote, N Swamy, and S Zanella-Béguelin. 2016. Short paper: Formal verification of smart contracts. In *Proceedings of the 11th ACM Workshop on Programming Languages and Analysis for Security (PLAS), in conjunction with ACM CCS*. 91–96.

Vitalik Buterin et al. 2014. A next-generation smart contract and decentralized application platform. *white paper* (2014).

Hao Chen, Xiongnan Newman Wu, Zhong Shao, Joshua Lockerman, and Ronghui Gu. 2016. Toward compositional verification of interruptible OS kernels and device drivers. In *ACM SIGPLAN Notices*, Vol. 51. ACM, 431–447.

Koen Claessen and John Hughes. 2011. QuickCheck: a lightweight tool for random testing of Haskell programs. *Acm sigplan notices* 46, 4 (2011), 53–64.

Cyril Cohen, Thierry Coquand, Simon Huber, and Anders Mörtberg. 2016. Cubical Type Theory: a constructive interpretation of the univalence axiom. *CoRR* abs/1611.02108 (2016). arXiv:1611.02108 http://arxiv.org/abs/1611.02108

Oded Goldreich, Silvio Micali, and Avi Wigderson. 1991. Proofs that yield nothing but their validity or all languages in NP have zero-knowledge proof systems. *Journal of the ACM (JACM)* 38, 3 (1991), 690–728.

Charles Antony Richard Hoare. 1978. Communicating sequential processes. In *The origin of concurrent programming*. Springer, 413–443.

Paul Hudak, Simon Peyton Jones, Philip Wadler, Brian Boutel, Jon Fairbairn, Joseph Fasel, María M Guzmán, Kevin Hammond, John Hughes, Thomas Johnsson, et al. 1992. Report on the programming language Haskell: a non-strict, purely functional language version 1.2. *ACM SigPlan notices* 27, 5 (1992), 1–164.

Atsushi Igarashi, Benjamin C Pierce, and Philip Wadler. 2001. Featherweight Java: a minimal core calculus for Java and GJ. *ACM Transactions on Programming Languages and Systems (TOPLAS)* 23, 3 (2001), 396–450.

Aggelos Kiayias, Alexander Russell, Bernardo David, and Roman Oliynykov. 2017. Ouroboros: A provably secure proof-of-stake blockchain protocol. In *Annual International Cryptology Conference*. Springer, 357–388.

Wen Kokke and Wouter Swierstra. 2015. Auto in agda. In *International Conference on Mathematics of Program Construction*. Springer, 276–301.

Per Martin-Löf and Giovanni Sambin. 1984. *Intuitionistic type theory*. Vol. 9. Bibliopolis Naples.

Satoshi Nakamoto. 2008. Bitcoin: A peer-to-peer electronic cash system. (2008).

Ulf Norell. 2008. Dependently typed programming in Agda. In *International School on Advanced Functional Programming*. Springer, 230–266.

Simon Peyton Jones, Jean-Marc Eber, and Julian Seward. 2000. Composing contracts: an adventure in financial engineering. *ACM SIG-PLAN Notices* 35, 9 (2000), 280–292.

Meni Rosenfeld. 2012. Overview of colored coins. *White paper, bitcoil. co. il* 41 (2012).

Pablo Lamela Seijas and Simon Thompson. 2018. Marlowe: Financial contracts on blockchain. In *International Symposium on Leveraging Applications of Formal Methods*. Springer, 356–375.

Pablo Lamela Seijas, Simon J Thompson, and Darryl McAdams. 2016. Scripting smart contracts for distributed ledger technology. *IACR Cryptology ePrint Archive* 2016 (2016), 1156.

Ilya Sergey, Amrit Kumar, and Aquinas Hobor. 2018. Scilla: a smart contract intermediate-level language. *arXiv preprint arXiv:1801.00687* (2018).

Anton Setzer. 2018. Modelling Bitcoin in Agda. *arXiv preprint arXiv:1804.06398* (2018).

Paul Van Der Walt and Wouter Swierstra. 2012. Engineering proof by reflection in Agda. In *Symposium on Implementation and Application of Functional Languages*. Springer, 157–173.

Joachim Zahnentferner. 2018. An Abstract Model of UTxO-based Cryptocurrencies with Scripts. *IACR Cryptology ePrint Archive* 2018 (2018), 469.

Joachim Zahnentferner. 2019. Multi-Currency Ledgers. (2019), To Appear.

Joachim Zahnentferner and Input Output HK. 2018. *Chimeric ledgers: Translating and unifying utxo-based and account-based cryptocurrencies*. Technical Report. Cryptology ePrint Archive, Report 2018/262, 2018. https://epri nt. iacr. org ….

… used throughout our formalization …

SECTION A

List Utilities

A.1 Indexed Operations

A.2 Inductive Relations

SECTION B

Set-like Interface for Lists

B.1 Decidable equality

B.2 Set Operations

SECTION C

Generalized Variables

We (ab)use Agda's recent capabilities for *generalized variables*, which allow one to declare variable names of a certain type at the top-level and then omit them from their usage in type definitions for clarity.

Below we give a complete set of all variables used throughout this thesis:

variable

$ads\ ads'\ ads''\ rads\ ads^r\ rads^r\ ads^l\ rads^l : AdvertisedContracts$

$cs\ cs'\ cs''\ rcs\ cs^r\ rcs^r \qquad cs^l\ rcs^l : ActiveContracts$

$ds\ ds'\ ds''\ rds\ ds^r\ rds^r \qquad ds^l\ rds^l : Deposits$

$\Gamma_0 : Configuration\ ads\ cs\ ds$

$\Gamma' : Configuration\ ads'\ cs'\ ds'$

$\Gamma'' : Configuration\ ads''\ cs''\ ds''$

$p_1\ p_1' : AdvertisedContracts \times AdvertisedContracts$

$p_2\ p_2' : ActiveContracts \times ActiveContracts$

$p_3\ p_3' : Deposits \times Deposits$

$p : Configuration'\ p_1\ p_2\ p_3$

$p' : Configuration'\ p_1'\ p_2'\ p_3'$

List of Figures

List of Tables